# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Arch Haskell Team <arch-haskell@haskell.org>

_hkgname=feed
pkgname=haskell-feed
pkgver=1.3.2.1
pkgrel=136
pkgdesc="Interfacing with RSS (v 0.9x, 2.x, 1.0) + Atom feeds."
url="https://github.com/haskell-party/feed"
license=("BSD")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-base-compat' 'haskell-old-locale' 'haskell-old-time' 'haskell-safe'
         'haskell-time-locale-compat' 'haskell-utf8-string' 'haskell-xml-types'
         'haskell-xml-conduit')
makedepends=('ghc' 'haskell-doctest' 'haskell-doctest-driver-gen' 'haskell-hunit'
             'haskell-test-framework' 'haskell-test-framework-hunit' 'haskell-markdown-unlit'
             'haskell-syb')
source=("https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz")
sha512sums=('2175201ff9eaab5732e1c4e8dda6a35e8fa0dea5c58cd379348c97cd61a5eaf76e88f8d58d22df0a4b16681017e7ad01657ad60a63d90dab88df3b644ccb1ea0')

build() {
    cd $_hkgname-$pkgver

    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test || echo "doctests are broken for now"
}

package() {
    cd $_hkgname-$pkgver

    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
